Example Technology Exercise 13 James Jones

Do some setup
Load the vect utility file so we have grad command.
Define gradient() function to go ahead and evaluate
the grad() function.

```
(%i1) load("vect")$
      gradient(f):=ev(express(grad(f)),diff)$
      norm(u):=sqrt(u.u)$
```

1 See Word Document

2 Problem 13.8.19

Define function

```
(%i4) f:(x^2+4*y^2)*exp(1-x^2-y^2)-z;
```

(%o4) $\left(4\,y^2+x^2\right)\,\%e^{-y^2-x^2+1}-z$

Find the first partials

```
(%i5) fx:diff(f,x),factor;
      fy:diff(f,y),factor;
```

(%o5) $-2\,x\,\left(4\,y^2+x^2-1\right)\,\%e^{-y^2-x^2+1}$

(%o6) $-2\,y\,\left(4\,y^2+x^2-4\right)\,\%e^{-y^2-x^2+1}$

Since the exp(1-x^2-y^2) can never be zero,
we focus on the rest

```
(%i7) fx1:fx/(exp(1-x^2-y^2));
      fy1:fy/(exp(1-x^2-y^2));
```

(%o7) $-2\,x\,\left(4\,y^2+x^2-1\right)$

(%o8) $-2\,y\,\left(4\,y^2+x^2-4\right)$

(%i9) sol:solve([fx1,fy1],[x,y]);

(%o9) $[[x=0,y=0],[x=-1,y=0],[x=1,y=0],[x=0,y=-1],[x=0,y=1]]$

We have critical points at (0,0), (-1,0), (1,0), (0,-1), and (0,1)

Find the second partials
and d to be fxx*fyy - fxy^2

```
(%i10) fxx:diff(fx,x),factor;
       fxy:diff(fx,y),factor;
       fyx:diff(fy,x),factor;
       fyy:diff(fy,y),factor;
```

(%o10) $2\left(8x^2y^2-4y^2+2x^4-5x^2+1\right)\%e^{-y^2-x^2+1}$

(%o11) $4xy\left(4y^2+x^2-5\right)\%e^{-y^2-x^2+1}$

(%o12) $4xy\left(4y^2+x^2-5\right)\%e^{-y^2-x^2+1}$

(%o13) $2\left(8y^4+2x^2y^2-20y^2-x^2+4\right)\%e^{-y^2-x^2+1}$

(%i14) d:fxx*fyy-fxy^2;

(%o14) $4\left(8x^2y^2-4y^2+2x^4-5x^2+1\right)\left(8y^4+2x^2y^2-20y^2-x^2+4\right)\%e^{-2y^2-2x^2+2}-16x^2y^2\left(4y^2+x^2-5\right)^2\%e^{2\left(-y^2-x^2+1\right)}$

Now plug the critical points into d

```
(%i15) makelist(subst(sol[k],d),k,1,5);
       makelist(subst(sol[k],fxx),k,1,5);
```

(%o15) $[16\%e^2,-24,-24,96,96]$

(%o16) $[2\%e,-4,-4,-6,-6]$

```
At (0,0),  d = 16e^2 > 0 and fxx = 2e > 0, so it is a relative minimum
At (-1,0), d = -24   < 0, so it is a saddle point
At (1,0),  d = -24   < 0, so it is a saddle point
At (0,-1), d = 96    > 0 and fxx = -6 < 0, so it is a relative maximum
At (0,1),  d = 96    > 0 and fxx = -6 < 0, so it is a relative maximum
```

3 Problem 13.7.29

Define the problem

```
(%i17) f:x*y^2+3*x-z^2-8;
       [x0,y0,z0]:[1,-3,2];
```

(%o17) $-z^2+x\,y^2+3\,x-8$

(%o18) $[1,-3,2]$

Find the gradient

```
(%i19) delf:gradient(f);
       n:subst([x=x0,y=y0,z=z0],delf);
```

(%o19) $[y^2+3,2\,x\,y,-2\,z]$

(%o20) $[12,-6,-4]$

3.1 Find a unit normal vector

```
(%i21) n/norm(n);
```

(%o21) $[\frac{6}{7},-\frac{3}{7},-\frac{2}{7}]$

3.2 Find the an equation of the tangent plane

```
(%i22) n.[x-x0,y-y0,z-z0]=0,expand;
```

(%o22) $-4\,z-6\,y+12\,x-22=0$

3.3 Find symmetric equations of normal line

These are equal to each other.

```
(%i23) [(x-x0)/n[1],(y-y0)/n[2],(z-z0)/n[3]];
```

(%o23) $[\frac{x-1}{12},-\frac{y+3}{6},-\frac{z-2}{4}]$

4 Problem 13.10.17

```
(%i24) f:x*y*z;
       g1:x+y+z-32;
       g2:x-y+z;
```

(%o24) $x\,y\,z$

(%o25) $z+y+x-32$

(%o26) $z-y+x$

```
Although we could write the equation out with an
equal sign, that makes it harder for Maxima to work with
```

```
(%i27) lside:gradient(f);
       rside:k*gradient(g1)+m*gradient(g2);
```

(%o27) $[y\,z,x\,z,x\,y]$

(%o28) $[m+k,k-m,m+k]$

```
Solve a system of equations
There are three from the objective function
plus the two constraints
```

```
(%i29) solve([lside[1]=rside[1],lside[2]=rside[2],lside[3]=rside[3],g1=0,g2=0]);
```

(%o29) $[[x=8,m=32,k=96,z=8,y=16]]$

```
Substitute these values into the objective
```

```
(%i30) subst(%,f);
```

(%o30) 1024

```
The maximum is 1024 when x = 8, y = 16, z = 8
```